RP 197-2003
Revision of RP 197-1998

SMPTE RECOMMENDED PRACTICE

Film to Video Transfer List

1 Scope

1.1 This practice describes the data format of a list of film-to-video transfers. This list may be stored and referenced to document the process of transferring pictures and sound from film-related media to video media. The information documented in this list may include general production and post-production information (date, facility, film type). The list also contains the specific time (synchronization) relationships between source and destination elements. The storage medium for this list is not specified by this practice.

1.2 This practice is based on ANSI/SMPTE 258M. The edit decision list (EDL) forms the base description of the structure of this film transfer list (FTL). Sections of this practice describe additional information that is used to document film-to-tape transfer sessions. Only the differences between EDL and FTL are contained here.

2 Normative references

The following standards contain provisions which, through reference in this text, constitute provisions of this practice. At the time of publication, the editions indicated were valid. All standards are subject to revision, and parties to agreements based on this practice are encouraged to investigate the possibility of applying the most recent edition of the standards indicated below.

ANSI/SMPTE 258M-1993, Television — Transfer of Edit Decision Lists

SMPTE RP 194-1998, Film Negative Cutter's Conform List

3 Notation

3.1 Terms

Grammatical terms are capitalized, with an underscore separating words. Some terms are used in a specific and formal manner to identify the meaning of that term in the context of a film transfer list. This type of term is printed in upper case using underscores (_) to join words into one term. Each of these terms will be defined in a formal manner using Backus-Naur Form (BNF) or the defined terms of 3.2 and ANSI/SMPTE 258M.

3.2 Defined terms in addition to the SMPTE EDL

3.2.1 film: Used to include both picture and sound sources on film-related media.

3.2.2 video: Used to include both video and audio destinations of video-related recording formats.

Approved
October 20, 2003

4 Film transfer list — Overall structure

4.1 General information

The FTL replaces the EDL component in the grammar. Like the EDL, the FTL consists of the INITIAL_DIRECTIVES followed by the FILM_TRANSFER_LIST.

```
FTL:
        INITIAL_DIRECTIVES FILM_TRANSFER_LIST 'SUB'
      | INITIAL_DIRECTIVES FILM_TRANSFER_LIST
      ;
```

The FILM_TRANSFER_LIST replaces the EDIT_LIST grammar item in an FTL and is made up of ELEMENTs. The physical order of the ELEMENTS in the list is representative of the order in which the transfers occurred.

```
FILM_TRANSFER_LIST:
        ELEMENT
      | ELEMENT FILM_TRANSFER_LIST
      ;
```

4.2 Film transfer list elements

There are six types of elements in the list:

- system directive
- information element
- transfer element
- user element
- trigger element
- device control element

```
ELEMENT:
        [SEPARATOR] TERMINATOR
      | SYSTEM_DIRECTIVE_ELEMENT [SEPARATOR] TERMINATOR
      | INFORMATION_ELEMENT [SEPARATOR] TERMINATOR
      | TRANSFER_ELEMENT [SEPARATOR] TERMINATOR
      | USER_ELEMENT [SEPARATOR] TERMINATOR
      | TRIGGER_ELEMENT [SEPARATOR] TERMINATOR
      | DEVICE_CONTROL_ELEMENT [SEPARATOR] TERMINATOR
```

5 System directives

5.1 System directives of the film transfer list are not the same as the SYSTEM_DIRECTIVE of the EDL. They are used to contain information used by the transfer system to perform and document the transfer at the system level. This information has meaning across several transfer elements. When a system directive is encountered in the list, the effect of the directive remains until cancelled by another directive. Some system directives act as default values for fields within a TRANSFER_ELEMENT. When a new value is specified in a transfer, the transfer suspends the system directive's value for that transfer only.

```
SYSTEM_DIRECTIVE_ELEMENT:
        "*" PRINTABLE_STRING
      | "INCLUDE" SEPARATOR FILENAME "=" TIMECODE
      | "NORECORD" SEPARATOR MODE_FIELD SEPARATOR MEDIUM_IDENTIFIER
      | "RECORD" SEPARATOR MODE_FIELD SEPARATOR MEDIUM_IDENTIFIER ["=" TIMECODE]
      ;
```

5.2 '*' directive

The '*' directive is the same as the '*' item in the EDL's INFORMATIONAL_DIRECTIVE. It is a system level comment.

5.3 INCLUDE, NORECORD and RECORD directives

The INCLUDE, NORECORD, and RECORD directives are the same as the EDL's CONTROL_DIRECTIVEs of the same names.

6 Information elements

6.1 The INFORMATION_ELEMENT is unique to the FTL. It is used to create specific data items for the session. Data items are formed by an IDENTIFIER followed by the character ':'. Some specific INFORMATION_ELEMENT names are defined by this guideline. Other IDENTIFIERs followed by ':' can be created in the list as long as the name is different from those specified here. The content of the nonreserved names is the PRINTABLE_STRING that follows. The content of each reserved INFORMATION_ELEMENT name is as defined in this clause.

```
INFORMATION_ELEMENT:
        "FRAME_RATE:" SEPARATOR ROLL_ID SEPARATOR REAL_NUMBER
                [SEPARATOR REAL_NUMBER]
      | "RECORD_FRAME_RATE:" SEPARATOR REAL_NUMBER
      | "EDGE_NUMBER_PERIOD:" SEPARATOR ROLL_ID SEPARATOR INTEGER "/" INTEGER
      | "EDGE_NUMBER_TYPE:" SEPARATOR ROLL_ID SEPARATOR EDGE_NUMBER_TYPE
      | "FILM_FORMAT:" SEPARATOR ROLL_ID SEPARATOR FILM_FORMAT
      | "FILM_ASPECT_RATIO:" SEPARATOR ROLL_ID SEPARATOR ASPECT_RATIO
      | "FILM_APERTURE:" SEPARATOR ROLL_ID SEPARATOR FILM_APERTURE
      | "VIDEO_ASPECT_RATIO:" SEPARATOR VIDEO_ASPECT_RATIO
      | "VIDEO_IMAGE_UNITS:" SEPARATOR VIDEO_IMAGE_UNITS
      | "PAN:" SEPARATOR PAN_PARAMETERS
      | "ZOOM:" SEPARATOR ZOOM_PARAMETERS
      | "ROTATION:" SEPARATOR ROT_PARAMETERS
      | "OPERATOR:" SEPARATOR PRINTABLE_STRING
      | "TRANSFER_FACILITY:" SEPARATOR PRINTABLE_STRING
      | "PRODUCTION_NUMBER:" SEPARATOR PRINTABLE_STRING
      | "PRODUCTION_TITLE:" SEPARATOR PRINTABLE_STRING
      | "SHOOT_DATE:" SEPARATOR PRINTABLE_STRING
      | "TRANSFER_DATE:" SEPARATOR PRINTABLE_STRING
      | "SOURCE_MEDIUM:" SEPARATOR ROLL_ID SEPARATOR PRINTABLE_STRING
      | IDENTIFIER ':' SEPARATOR PRINTABLE_STRING
      ;
```

6.2 FRAME_RATE:

The "FRAME_RATE:" item is used to specify the natural frame rate and the transfer frame rate for each source.

The first value is expressed as a real number and should contain enough precision to reflect the exact frame rate of the source material when it was produced; however, it is often an integer (e.g., 24). The second value, if present, is also a real number and represents the frame rate to which the source was resolved during transfer. FRAME_RATE must be specified for each roll in a transfer session; there is no default.

In the case of a film source in an NTSC system, the first value might be "24," and the second value might be "23.98," reflecting the standard practice and field rate of NTSC. If 25 frames/s film or audio sources were

transferred in a 50-Hz system, the second value would not be needed to document a standard frame-per-frame transfer. The ratio between the two values documents variable speed transfer rates that may take place in a transfer. Although two digits of decimal precision are an approximation of the exact frequency involved, implementations may round to the closest exact frame rate on the system. If an ambiguity must be avoided, additional digits of decimal precision should be used.

6.3 RECORD_FRAME_RATE:

The "RECORD_FRAME_RATE" item is used to specify the exact frame rate of the device recording the transfer. It is expressed as a real number and should contain enough precision to reflect the exact frame rate of the device. For NTSC systems, the value is usually 29.97. RECORD_FRAME_RATE must be specified for each transfer session; there is no default.

6.4 EDGE_NUMBER_PERIOD:

"EDGE_NUMBER_PERIOD:" is used to note the number of film frames between whole edge numbers on the source medium identified. This may be used by the system to calculate conversions based on frame rates, time, and edge numbers. The value in EDGE_NUMBER_PERIOD is expressed as a ratio of two integers, the number of perforations per edge number, and the number of perforations per frame.

When the first integer is not divisible by the second integer (as in the example below), a "perf specifier" is required to identify unambiguously the film frame referred to by transfer elements. Refer to SMPTE RP 195 for a description of the perf specifier. EDGE_NUMBER_PERIOD must be specified for each roll in a transfer session; there is no default. Example:

```
EDGE_NUMBER_PERIOD: ROLL01 64/3
```

6.5 EDGE_NUMBER_TYPE:

"EDGE_NUMBER_TYPE:" records the method used to mark the edge numbers on the identified source roll.

```
EDGE_NUMBER_TYPE:
        EDGE_NUMBER_TYPE_ITEM [SEPARATOR EDGE_NUMBER_DIRECTION]
      | EDGE_NUMBER_TYPE [SEPARATOR] EDGE_NUMBER_TYPE_ITEM
            [SEPARATOR EDGE_NUMBER_DIRECTION]
      ;

EDGE_NUMBER_TYPE_ITEM:
        "BAR" /* machine readable bar code */
      | "LHR" /* human readable version of machine readable bar code*/
      | "LAT" /* latent image characters */
      | "INK" /* other printed characters */
      ;

EDGE_NUMBER_DIRECTION
        "INCR" / * incrementing edge numbers * /
      | "DECR" / * decrementing edge numbers; i.e., tails out * /
      ;
```

If no EDGE_NUMBER_DIRECTION is specified, INCR is assumed. Note that two varieties of edge number may be specified simultaneously, and that the count directions may be different. EDGE_NUMBER_TYPE must be specified for each roll in a transfer session; there is no default. For example:

```
EDGE_NUMBER_TYPE: ROLL01 BAR INCR INK DECR
```

6.6 FILM_FORMAT:

The "FILM_FORMAT:" data item is used to identify the specific gauge and variant of the transfer process for all film sources during the time the directive is active. There is no default value for this data item. (The defined contents of this data field are the same as those used in EBU Tech. 3245 Supplement 4.) FILM_FORMAT must be specified for each roll in a transfer session; there is no default.

```
FILM_FORMAT:
          "16MM"
        | "SUPER 16"
        | "35MM"
        | "35MM/3"
        | "35MM/4
        | "35MM/8"
        | "35MM/12"
        | "SUPER 35"
        | "65MM"
        | "65MM/5"
        | "65MM/8"
        | "65MM/10"
        | "65MM/15"
        | NUMBER "MM/" NUMBER
        ;
```

6.7 FILM_ASPECT_RATIO:

"FILM_ASPECT_RATIO:" expresses a ratio of width to height for the intended production aperture of the source images. The standard television aspect ratio of 4:3 (1.33:1) need not be specified and is considered the default. This data item is assumed to affect all film sources transferred during the time the particular directive is in force.

A film aspect ratio specified different from the default means that the rectangular area of that proportion bounded horizontally by the FILM_APERTURE on the film is completely contained in the video production aperture. Aspect ratios wider than 4:3 will just fill the horizontal area of the video production aperture with the height of the picture being less than the height of the aperture. Black fill for letterboxing is implied. FILM_ASPECT_RATIO may be specified for each roll in a transfer session; by default, it is 4:3.

The modifier "ANAMORPHIC" implies that the distorted film image is adjusted to normal proportions before fitting to the applicable aspect ratio.

```
ASPECT_RATIO:
          WIDTH ':' HEIGHT ["ANAMORPHIC"]
        ;

WIDTH:
          REAL_NUMBER
        ;

HEIGHT:
          REAL_NUMBER
        ;
```

6.8 FILM_APERTURE:

"FILM_APERTURE:" describes the production aperture of the source images. It is either a reference to the appropriate SMPTE standard or a PRINTABLE_STRING, which may be used to describe a non-SMPTE aperture.

FILM_APERTURE may be specified for each roll in a transfer session; the default for a given FILM_FORMAT is the SMPTE standard for the aspect ratio specified. In many cases, this would be SMPTE 96M.

```
FILM_APERTURE:
        "SMPTE" SEPARATOR "STD" NUMBER "M" ["--" NUMBER]
      | "SMPTE" SEPARATOR "RP" NUMBER ["--" NUMBER]
      | PRINTABLE_STRING
      ;
```

6.9 VIDEO_ASPECT_RATIO:

"VIDEO_ASPECT_RATIO:" expresses a ratio of width to height for the format of the production aperture in the video image output from the transfer. The standard television aspect ratio of 4:3 (1:33:1) need not be specified and is considered the default.

```
VIDEO_ASPECT_RATIO:
        WIDTH ":" HEIGHT
      ;
```

6.10 VIDEO_IMAGE_UNITS:

"VIDEO_IMAGE_UNITS:" specifies the number of units of measurement of image width and height in the video output format production aperture. Typical values might be 8 6 (corresponding to common practice in the user interfaces of image manipulation devices), or 100 100 (corresponding to measurements as a percentage of picture).

The default is 720 x 486 for TIME_CODE_MODULUS of 30 and 720 x 576 for TIME_CODE_MODULUS of 25. These correspond to the actual video raster of ANSI/SMPTE 125M and EBU Tech. 3267.

```
VIDEO_IMAGE_UNITS:
        HORIZONTAL_VIDEO_IMAGE_UNITS ':' VERTICAL_VIDEO_IMAGE_UNITS
      ;

HORIZONTAL_VIDEO_IMAGE_UNITS:
        NUMBER
      ;

VERTICAL_VIDEO_IMAGE_UNITS:
        NUMBER
      ;
```

Example:

```
VIDEO_IMAGE_UNITS: 24:18
```

6.11 PAN

"PAN" describes the amount of repositioning to frame an incompatible film image to the video production aperture in use. Any number of PAN elements may appear in an FTL; when a PAN has been specified, it

applies to all subsequent transfer elements until superseded by the next PAN element. After a PAN has been specified, panning may be reset to none using the element PAN 0 0.

The PAN element is optional; the default is no panning, where the center of the film production aperture corresponds exactly to the center of the video production aperture. This is equivalent to the element PAN 0 0. If no PAN element appears in an FTL, the default applies to the entire FTL. PAN may be applied in combination with ZOOM and ROTATION (see 6.12 and 6.13). When PAN is specified, it is to be applied to the image before any ZOOM or ROTATION. The syntax of a PAN element is as follows:

"PAN:" SEPARATOR PAN_PARAMETERS

PAN_PARAMETERS:
    PAN_UNITS SEPARATOR TILT_UNITS
;

PAN_UNITS:
    REAL_NUMBER
;

TILT_UNITS:
    REAL_NUMBER
;

For example:

PAN 4 1.6

Horizontal position is specified by PAN_UNITS as a number of HORIZONTAL_VIDEO_IMAGE_UNITS. Although arbitrary precision may be specified, accuracy and repeatability better than 1/16 pixel are not anticipated. The parameter is signed, specifying the shift in the center of the aperture. Positive count implies a rightward shift of the video production aperture relative to the film aperture, which results in a leftward shift of the subject material within the output image.

Vertical position is specified by TILT_UNITS as a number of VERTICAL_VIDEO_IMAGE_UNITS. Although arbitrary precision may be specified, accuracy and repeatability better than 1/16 line are not anticipated. The parameter is signed, specifying the shift in the center of the aperture. Positive count implies a downward shift of the video production aperture relative to the film aperture, which results in an upward shift of the subject material within the output image.

Pan = tilt = 0 represents the default positioning (center mapped to center). Both horizontal and vertical position are specified in absolute terms, relative to the default positioning. Therefore, the position specified by each PAN element is absolute and does not depend on any other PAN element. For example, assuming the following directives are in force:

FILM_ASPECT_RATIO: 16:9
VIDEO_ASPECT_RATIO: 4:3
VIDEO_IMAGE_UNITS: 24:18
PAN -2 0

The relation of the video production aperture to the film aperture is as illustrated in figure 1.



Figure 1 – Relation of pan video production aperture to film aperture

6.12 ZOOM

"ZOOM:" describes the amount of resizing (zooming and cropping) to frame an incompatible film image to the video production aperture in use. Any number of ZOOM elements may appear in an FTL; when a ZOOM has been specified, it applies to all subsequent transfer elements until superseded by the next ZOOM element. After a ZOOM has been specified, zooming may be reset to none using the element ZOOM 1 1.

The ZOOM element is optional; the default is that the rectangular area of the film specified by the FILM_ASPECT_RATIO, FILM_FORMAT, and FILM_APERTURE is just contained within the rectangular area specified by the VIDEO_ASPECT_RATIO, with the proviso that film aspect ratios wider than 4:3 will be letterboxed — they will just fill the horizontal area of the video production aperture, with the height of the picture being less than the height of the production aperture (it is implied that the unused vertical area is filled with black). This is equivalent to the element ZOOM 1. If no ZOOM element appears in an FTL, the default applies to the entire FTL.

ZOOM may be applied in combination with PAN and ROTATION (see 6.11 and 6.13). When ZOOM is specified, it is to be applied to the image after any PAN and before any ROTATION. The syntax of a ZOOM element is as follows:

"ZOOM:" SEPARATOR ZOOM_PARAMETERS

ZOOM_PARAMETERS:
    ZOOM_HORIZONTAL [SEPARATOR ZOOM_VERTICAL]

ZOOM_HORIZONTAL:
    REAL_NUMBER
  ;

ZOOM_VERTICAL:
    REAL_NUMBER
  ;

For example:

ZOOM 1.25 1.25

ZOOM_PARAMETERS specified different from the default means that the rectangular area of that size and proportion on the film is completely contained in the video production aperture. Parameters are specified as multiplicative factors on the image dimensions, from the scanned film area to the video production aperture.

The default zoom is such that film aspect ratios wider than 4:3 will just fill the horizontal area of the video production aperture with the height of the picture being less than the height of the production aperture (it is implied that the unused vertical area is filled with black). Thus, a default virtual film scan size is implied whose width is the number of HORIZONTAL_VIDEO_IMAGE_UNITS in the film aperture for the default zoom and whose height is the number of VERTICAL_VIDEO_IMAGE_UNITS in the actual film aperture for no aspect ratio distortion. ZOOM_PARAMETERS specify a substitute scan in terms of the virtual scan.

Horizontal zoom is specified by ZOOM_HORIZONTAL as a multiplicative factor on the horizontal scanned image extent, relative to the default, where numbers greater than 1 imply zooming in (that is, scanning of a smaller extent to produce the full width of the video production aperture). Although arbitrary precision may be specified, accuracy and repeatability better than 1/16 pixel are not anticipated. The parameter is signed. A negative count implies horizontal reflection by backward scanning.

Vertical zoom is specified by ZOOM_VERTICAL as a multiplicative factor on the vertical scanned image extent, relative to the default, where numbers greater than 1 imply zooming in (that is, scanning of a smaller extent to produce the full height of the video production aperture). Although arbitrary precision may be specified, accuracy and repeatability better than 1/16 line are not anticipated. The parameter is signed. A negative count implies vertical reflection by backward scanning. If ZOOM_VERTICAL is not specified, it is assumed to be calculated from ZOOM_HORIZONTAL for no aspect ratio distortion.

ZOOM 1 1 represents the default zoom (no resizing). Both horizontal and vertical zoom are specified in absolute terms, relative to the default. Therefore, the resizing specified by each ZOOM element is absolute and does not depend on any other ZOOM element. Continuing the example of PAN, if the following directives are in force:

FILM_ASPECT_RATIO: 16:9
VIDEO_ASPECT_RATIO: 4:3
VIDEO_IMAGE_UNITS: 24:18

PAN -2 0
ZOOM 1.25

The relation of the video production aperture to the film aperture is as illustrated in figure 2.



Figure 2 – Relation of zoom video production aperture to film aperture

6.13 ROTATION

"ROTATION:" describes the amount of rotation of the video production aperture relative to the film aperture. Any number of ROTATION elements may appear in an FTL; when a ROTATION has been specified, it applies to all subsequent transfer elements until superseded by the next ROTATION element. After a ROTATION has been specified, rotation may be reset to none using the element ROTATION 0. The ROTATION element is optional; the default is no rotation. This is equivalent to the element ROTATION 0. If no ROTATION element appears in an FTL, the default applies to the entire FTL.

Rotation may be applied in combination with PAN and ZOOM (see 6.11 and 6.12). When ROTATION is specified, it is to be applied to the image after any PAN or ZOOM. The syntax of a ROTATION element is as follows:

"ROTATION:" SEPARATOR ROT_PARAMETERS

ROT_PARAMETERS:
    ROT_ANGLE
  ;

ROT_ANGLE:
    REAL_NUMBER
  ;

For example:

ROTATION –90

The angle of rotation ROT_ANGLE is specified in degrees. A positive number implies that the image sensor is rotated counterclockwise relative to the film. Although arbitrary precision may be specified, accuracy and repeatability better than 1/16 line or pixel at the extreme edges of the scanned image are not anticipated.

The center of rotation is the center of the image specified by the PAN element which is in force. An effect equivalent to rotation about a point which is not the center of the video production aperture may be specified by adjustment of the PAN parameters. Continuing the example of PAN and ZOOM, if the following directives are in force:

FILM_ASPECT_RATIO: 16:9
VIDEO_ASPECT_RATIO: 4:3
VIDEO_IMAGE_UNITS: 24:18

PAN -2 0
ZOOM 1.25
ROTATION 5

The relation of the video production aperture to the film aperture is as illustrated in figure 3:



Figure 3 – Relation of rotation video production aperture to film aperture

6.14 OPERATOR: and TRANSFER_FACILITY:

"OPERATOR:" and "TRANSFER_FACILITY:" provide standard items for a printable string to indicate the source of the transfers in the list.

6.15 PRODUCTION_NUMBER: and PRODUCTION_TITLE:

"PRODUCTION_NUMBER:" and "PRODUCTION_TITLE:" provide standard identifiers for a printable string to be determined by the operator.

6.16 SHOOT_DATE: and TRANSFER_DATE:

"SHOOT_DATE:" and "TRANSFER_DATE:" items record day, month, and year as determined by the transfer operator. The date is recorded as a PRINTABLE_STRING. It is expected that this will be parsed according to local preferences.

6.17 SOURCE_MEDIUM:

The "SOURCE_MEDIUM:" data item is used to associate a printable string with a ROLL_ID used to refer to that source in the list. For example, a source roll of film footage may be identified as: Source_Medium: BKGND1 XYZ Stock Shot Catalog #1A-1596.

7 Transfer elements

7.1 The TRANSFER_ELEMENT replaces the EDL's SOURCE_ELEMENTs which are not allowed in an FTL. The TRANSFER_ELEMENT documents time and content relationships between a film source and the video destination format. Multiple adjacent TRANSFER_ELEMENTS (with the same TRANSFER_NUMBER) may be used to indicate simultaneous transfers from multiple sources, such as separate film and sound rolls.

```
TRANSFER_ELEMENT:
        TRANSFER_NUMBER SEPARATOR
        SOURCE_IDENTIFICATION_FIELD SEPARATOR
        TRANSFER_MODE SEPARATOR
        "T" SEPARATOR
        FILM_POSITION [ FILM_FRAME_TYPE ] SEPARATOR
        VIDEO_TIMECODE SEPARATOR
        TRANSFER_DURATION
      ;
```

7.2 TRANSFER_NUMBER

The TRANSFER_NUMBER field consists of one to six numeric characters, followed by an optional alphabetic character. The range of valid transfer numbers is 1 to 999999z. Leading zero characters are not significant and the number zero (0) is prohibited.

```
TRANSFER_NUMBER: /* limit of 6 digits */
        DIGIT
      | DIGIT ALPHA / * number zero is prohibited */
      ;
```

A TRANSFER_NUMBER that is equal to a TRANSFER_NUMBER in another TRANSFER_ELEMENT indicates that the transfers were performed simultaneously.

7.3 Source identification

The transfer's source is noted in the transfer by a roll identifier and an optional subfield to contain source audio information. The SOURCE_AUDIO_SUBFIELD is the same as the EDL.

```
SOURCE_IDENTIFICATION_FIELD:
        ROLL_ID
      | ROLL_ID '.' SOURCE_AUDIO_SUBFIELD
```

7.3.1 ROLL_ID

The ROLL_ID field consists of one to eight alphanumeric characters. It is an identifier for the source medium for this transfer.

7.3.2 Audio source information

The FTL supports source audio track selection and mixing in the same way the EDL supports audio sources.

7.4 TRANSFER_MODE

The TRANSFER_MODE field of a TRANSFER_ELEMENT is the same as the EDL's MODE_FIELD. This includes the same audio destination capacity by using the MODE_AUDIO_SUBFIELD.

7.5 Film position

This field specifies the first film frame of the transfer. The FILM_POSITION field consists of one or a sequence of subfields: FILM_FEET_EDGE_NUMBERs and FILM_TIMECODE. If more than one is specified, the first subfield is followed by an equal-sign character ('=') and then the second subfield with no other characters separating the two, and similarly for a third or subsequent subfield. The sequence of multiple film positions must follow the sequence of the separately-specified EDGE_NUMBER_TYPE. For transfers at the nominal 24 frames per second from film to the nominal 30 frames-per-second video, the optional subfield FILM_FRAME_TYPE is appended to the FILM_POSITION.

```
FILM_POSITION:
          FILM_EDGE_POSITION ["=" FILM_TIMECODE [FILM_FRAME_TYPE] ]
        | FILM_TIMECODE [FILM_FRAME_TYPE]
        ;

FILM_EDGE_POSITION:
          FILM_FEET_EDGE_NUMBER [FILM_FRAME_TYPE]
        | FILM_EDGE_POSITION [FILM_FEET_EDGE_NUMBER [FILM_FRAME_TYPE] ]
        ;
```

7.5.1 FILM_FEET_EDGE_NUMBER

This subfield may be a notation of a latent-image key number or of a printed or latent-image edge number, as specified by the EDGE_NUMBER_TYPE information element.

A latent image key-number notation consists of a numeric or alpha prefix, followed by 6 numeric digits, followed by a 4-digit KEY_FEET footage count, a plus character ("+"), and a two-digit numeric FILM_FRAME field. Arithmetic carries do not propagate from the KEY_FEET field. Additionally, a specifier for the unambiguous identification of the length of the film foot is appended, when the film format requires one. The "perf specifier" is described in SMPTE RP 195.

A printed or latent image edge number consists of an optional 8-character alphanumeric FILM_BATCH_NUMBER identifier or alpha prefix followed by a numeric FILM_FEET field, a plus character ("+"), and a two-digit numeric FILM_FRAME field. The allowable length of the FILM_FEET field varies, depending on the EDGE-NUMBER_TYPE. Arithmetic carries do not propagate from the FILM_FEET field to the FILM_BATCH_NUMBER field.

Note that the film transfer list allows only positive frame offsets from the start of foot. This is in contrast to, for example, a printed negative cutter's conform list (SMPTE RP 194), where personal taste of the human user may dictate positive or negative offsets.

```
FILM_FEET_EDGE_NUMBER:
        FILM_FEET_BAR_NUMBER
      | FILM_FEET_LHR_NUMBER
      | FILM_FEET_LAT_NUMBER
      | FILM_FEET_INK_NUMBER
      ;

FILM_FEET_BAR_NUMBER:
        {DIGIT}4 SEPARATOR KEY_NUM PLUS_MINUS FILM_FRAME [PERF_SPEC]

FILM_FEET_LHR_NUMBER:
        {ALPHA}2 SEPARATOR KEY_NUM PLUS_MINUS FILM_FRAME [PERF_SPEC]

KEY_NUM:
        {DIGIT}2 SEPARATOR {DIGIT}4 SEPARATOR KEY_FEET
      ;

KEY_FEET:
        {DIGIT}4
      ;

PLUS:
        "+"
      ;

PERF_SPEC:
        [SEPARATOR] "P" [DIGIT] DIGIT
      ;

FILM_FEET_LAT_NUMBER:
        [FILM_BATCH_NUMBER SEPARATOR] FILM_FEET PLUS FILM_FRAME
      ;

FILM_FEET_INK_NUMBER:
        [FILM_BATCH_NUMBER SEPARATOR] FILM_FEET PLUS FILM_FRAME
      ;

FILM_BATCH_NUMBER:
        {FILM_BATCH_NUMBER_CHAR}1..8
      ;

FILM_BATCH_NUMBER_CHAR:
        ALPHA
      | DIGIT
      ;

FILM_FEET:
        {DIGIT}1..6
      ;

FILM_FRAME:
        {DIGIT}2
      ;

Example:
      XA 43 4247 1932+04 P1
```

7.5.2 FILM_TIMECODE

This subfield consists of a time code number which represents the first film frame of the transfer element. Note that film oriented time code numbers do not make use of the field precision of time code notation.

7.5.3 FILM_FRAME_TYPE

When 24 frame-per-second film is transferred to 30 frame-per-second video, a technique commonly referred to as 3:2 pulldown is used. In such a transfer, each film frame is recorded on either 2 or 3 consecutive video fields. This creates a repeated sequence of four film frames recorded onto five video frames. In figure 4, the film frames are labeled A-D, and the video fields are labeled 1 and 2.



Figure 4 – Film frames and video fields

As figure 4 shows, when 24 frame-per-second film is transferred to 30 frame-per-second video tape, there are four ways in which a film frame is transferred, only one of which results in one film frame being transferred to a single video frame. The letters A, B, C, and D are used to designate which of the four different types of transfers is used for that frame.

Frame A is the only film frame which corresponds to a single video frame. Frame B is transferred to fields 1 and 2 of a video frame, plus field 1 of the following frame. Film frame C is transferred to a video field 2 plus field 1 of the following video frame. Film frame D is transferred to a video field 2 plus both fields of the following frame.

When 30 frame/s film is transferred frame for frame to video in a 30 frame/s system (that is, without pulldown), or 24 or 25 frame/s film is transferred in a 25 frame/s system, there are two possible film frame to video field relationships. The case where one film frame is transferred to video field 1 followed by field 2 is the same as frame type "A." The other relationship (field 2 followed by field 1 of the next video frame) is the same as frame type "C." These frame type indications may be used to document these relationships.

When 24 frame/s film is transferred in a 25 frame/s system at original speed, the situation is more complicated still. Here, typically, there are eleven "A" frames, followed by one "B" frame followed by eleven "C" frames followed by one "D" frame. In this case, the exact "A" or "C" frame transferred is indicated by appending an integer: 1 represents the first possible film frame of the given type; 11 represents the last possible film frame of that type. If the integer is omitted, the first possible film frame is implied.

Finally, in abnormal cases, it may be desirable to start a transfer from the second occurrence of field 1 in a "B" frame, or the second field 2 in a "D" frame, in either 25 frame/s or 30 frame/s systems. In these cases, the digit "2" is appended to the FILM_FRAME_TYPE.

```
FILM_FRAME_TYPE:
          'A' [NUMBER]
        | 'B' ['2']
        | 'C' [NUMBER]
        | 'D' ['2']
        ;
```

7.6 VIDEO_TIMECODE

This field consists of a time code number (with field precision) which represents the first video field containing information from the FILM_TIMECODE or FILM_FEET_EDGE_NUMBER source frame.

7.7 TRANSFER_DURATION

This field consists of a time code number (with field precision) or an integer, and indicates the total number of destination frames recorded with information from the source. If a time code number, it is usually a "field 2" number (with a colon or semicolon) to signify that a whole number of frames are transferred.

8 User element

The EDL's USER_ELEMENT is allowed in the FTL.

9 Trigger element

The EDL's TRIGGER_ELEMENT is allowed in the FTL.

10 Device control element

The EDL's DEVICE_CONTROL_ELEMENT is allowed in the FTL.

Annex A (normative)
Grammar terms

```
ASPECT_RATIO:
          WIDTH ':' HEIGHT [ "ANAMORPHIC" ]
        ;

EDGE_NUMBER_TYPE:
          EDGE_NUMBER_TYPE_ITEM [SEPARATOR EDGE_NUMBER_DIRECTION
        | EDGE_NUMBER_TYPE [SEPARATOR] EDGE_NUMBER_TYPE_ITEM [SEPARATOR
              EDGE_NUMBER_DIRECTION]
        ;

EDGE_NUMBER_TYPE_ITEM:
          "BAR"          /* machine readable bar code */
        | "LHR"          /* human readable version of machine readable bar code */
        | "LAT"          /* latent image characters */
        | "INK"          /* other printed characters */
        ;

EDGE_NUMBER_DIRECTION:
        "INCR"             / * incrementing edge numbers */
        | "DECR"           / * decrementing edge numbers; i.e., tails out */
        ;
```

```
ELEMENT:
          [SEPARATOR] TERMINATOR
        | SYSTEM_DIRECTIVE_ELEMENT [ SEPARATOR ] TERMINATOR
        | INFORMATION_ELEMENT [ SEPARATOR ] TERMINATOR
        | TRANSFER_ELEMENT [ SEPARATOR ] TERMINATOR
        | USER_ELEMENT [ SEPARATOR ] TERMINATOR
        | TRIGGER_ELEMENT [ SEPARATOR ] TERMINATOR
        | DEVICE_CONTROL_ELEMENT [ SEPARATOR ] TERMINATOR
        ;

ELEMENT_LIST:
          ELEMENT
        | ELEMENT_LIST ELEMENT
        ;

FILM_APERTURE:
          "SMPTE" SEPARATOR "STD" NUMBER "M" ["-" NUMBER]
        | "SMPTE" SEPARATOR "RP" NUMBER ["-" NUMBER]
        | PRINTABLE_STRING
        ;

FILM_FEET_EDGE_NUMBER:
          FILM_FEET_BAR_NUMBER
        | FILM_FEET_LHR_NUMBER
        | FILM_FEET_LAT_NUMBER
        | FILM_FEET_INK_NUMBER
        ;

FILM_FEET_BAR_NUMBER:
          {DIGIT}4 SEPARATOR KEY_NUM PLUS_MINUS FILM_FRAME [PERF_SPEC]
        ;

FILM_FEET_LHR_NUMBER:
          {ALPHA}2 SEPARATOR KEY_NUM PLUS_MINUS FILM_FRAME [PERF_SPEC]
        ;

FILM_FEET_LAT_NUMBER:
          [FILM_BATCH_NUMBER SEPARATOR] FILM_FEET PLUS FILM_FRAME
        ;

FILM_FEET_INK_NUMBER:
          [FILM_BATCH_NUMBER SEPARATOR] FILM_FEET PLUS FILM_FRAME
        ;

FILM_BATCH_NUMBER:
          {FILM_BATCH_NUMBER_CHAR}1..8
        ;

FILM_BATCH_NUMBER_CHAR:
          ALPHA
        | DIGIT
        ;

FILM_FEET:
          {DIGIT}1..6
        ;

FILM_FRAME:
          {DIGIT}2
        ;
```

```
FILE_FORMAT:
            "16MM"
          | “SUPER 16"
          | "35MM"
          | "35MM/3"
          | "35MM/4"
          | "35MM/8"
          | "35MM/12"
          | "SUPER 35"
          | "65MM"
          | "65MM/5"
          | "65MM/8"
          | "65MM/10"
          | "65MM/15"
          | NUMBER "MM/" NUMBER
          ;

FILM_FRAME_TYPE:
            ’A’ [NUMBER]
          | ’B’ [’2’]
          | ’C’ [NUMBER]
          | ’D’ [’2’]
          ;

FILM_POSITION:
            FILM_EDGE_POSITION ["=" FILM_TIMECODE [ FILM_FRAME_TYPE ] ]
          | FILM_TIMECODE [ FILM_FRAME_TYPE ]
          ;

FILM_EDGE_POSITION:
            FILM_FEET_EDGE_NUMBER [ FILM_FRAME_TYPE ]
          | FILM_EDGE_POSITION [ FILM_FEET_EDGE_NUMBER [ FILM_FRAME_TYPE ] ]
          ;

FILM_TIMECODE:
            TIMECODE
          ;

FILM_TRANSFER_LIST:
            ELEMENT
          | ELEMENT FILM_TRANSFER_LIST
          ;

FTL:
            INITIAL_DIRECTIVES FILM_TRANSFER_LIST ’SUB’
          | INITIAL_DIRECTIVES FILM_TRANSFER_LIST
          ;

HEIGHT:
            REAL_NUMBER
          ;

HORIZONTAL_VIDEO_IMAGE_UNITS:
            NUMBER
          ;

INFORMATION_ELEMENT:
            "FRAME_RATE:" SEPARATOR ROLL_ID SEPARATOR REAL_NUMBER [ SEPARATOR REAL_NUMBER ]
          | "RECORD_FRAME_RATE:" SEPARATOR REAL_NUMBER
          | "EDGE_NUMBER_PERIOD:" SEPARATOR ROLL_ID SEPARATOR INTEGER "/" INTEGER
          | "EDGE_NUMBER_TYPE:" SEPARATOR ROLL_ID SEPARATOR EDGE_NUMBER_TYPE
          | "FILM_FORMAT:" SEPARATOR ROLL_ID SEPARATOR FILM_FORMAT
```

```
        | "FILM_ASPECT_RATIO:" SEPARATOR ROLL_ID SEPARATOR ASPECT_RATIO
        | "FILM_APERTURE:" SEPARATOR ROLL_ID SEPARATOR FILM_APERTURE
        | "VIDEO_ASPECT_RATIO:" SEPARATOR VIDEO_ASPECT_RATIO
        | "VIDEO_IMAGE_UNITS:" SEPARATOR VIDEO_IMAGE_UNITS
        | "PAN:" SEPARATOR PAN_PARAMETERS
        | "ZOOM:" SEPARATOR ZOOM_PARAMETERS
        | "ROTATION:" SEPARATOR ROT_PARAMETERS
        | "OPERATOR:" SEPARATOR PRINTABLE_STRING
        | "TRANSFER_FACILITY:" SEPARATOR PRINTABLE_STRING
        | "PRODUCTION_NUMBER:" SEPARATOR PRINTABLE_STRING
        | "PRODUCTION_TITLE:" SEPARATOR PRINTABLE_STRING
        | "SHOOT_DATE:" SEPARATOR PRINTABLE_STRING
        | "TRANSFER_DATE:" SEPARATOR PRINTABLE_STRING
        | "SOURCE_MEDIUM:" SEPARATOR ROLL_ID SEPARATOR PRINTABLE_STRING
        | IDENTIFIER ':' SEPARATOR PRINTABLE_STRING
        ;

KEY_FEET:
          {DIGIT}4
;
KEY_NUM:
          {DIGIT}2 SEPARATOR {DIGIT}4 SEPARATOR KEY_FEET
        ;

PAN_PARAMETERS:
          PAN_UNITS SEPARATOR TILT_UNITS
        ;

PAN_UNITS:
          REAL_NUMBER
        ;

PERF_SPEC:
          [SEPARATOR] "P" [DIGIT] DIGIT
        ;

PLUS:
          "+"
        ;

ROT_PARAMETERS:
          ROT_ANGLE
        ;

ROT_ANGLE:
          REAL_NUMBER
        ;

SOURCE_IDENTIFICATION:
          ROLL_ID
        | ROLL_ID '.' SOURCE_AUDIO_SUBFIELD
        ;

SYSTEM_DIRECTIVE_ELEMENT:
          "*" PRINTABLE_STRING
        | "INCLUDE" SEPARATOR FILENAME "=" TIMECODE
        | "NORECORD" SEPARATOR MODE_FIELD SEPARATOR MEDIUM_IDENTIFIER
        | "RECORD" SEPARATOR MODE_FIELD SEPARATOR MEDIUM_IDENTIFIER [ "=" TIMECODE ]
        ;
```

```
TILT_UNITS:
        REAL_NUMBER
      ;

TRANSFER_DURATION:
        TIMECODE
      | NUMBER
      ;

TRANSFER_ELEMENT:
        TRANSFER_NUMBER SEPARATOR
        SOURCE_IDENTIFICATION SEPARATOR
        TRANSFER_MODE SEPARATOR
        "T" SEPARATOR
        FILM_POSITION [ FILM_FRAME_TYPE ] SEPARATOR
        VIDEO_TIMECODE SEPARATOR
        TRANSFER_DURATION
      ;

TRANSFER_MODE:
        'A'
      | 'V'
      | 'B'
      | 'A' MODE_AUDIO_SUBFIELD
      | 'B' MODE_AUDIO_SUBFIELD
      ;

TRANSFER_NUMBER: /* limit of 6 digits */
        DIGIT
      | DIGIT ALPHA /* number zero is prohibited */
      ;

VERTICAL_VIDEO_IMAGE_UNITS:
        NUMBER
      ;

VIDEO_ASPECT_RATIO:
        WIDTH ':' HEIGHT
      ;

VIDEO_IMAGE_UNITS:
        HORIZONTAL_VIDEO_IMAGE_UNITS ':' VERTICAL_VIDEO_IMAGE_UNITS
      ;

VIDEO_TIMECODE:
        TIMECODE
      ;

WIDTH:
        REAL_NUMBER
      ;

ZOOM_PARAMETERS:
        ZOOM_HORIZONTAL [SEPARTOR ZOOM_VERTICAL]
      ;

ZOOM_HORIZONTAL:
        REAL_NUMBER
      ;
```

```
ZOOM_VERTICAL:
        REAL_NUMBER
      ;
```

Annex B (informative)
Example 1

```
SMPTE RP 197
*This film transfer list conforms to the SMPTE FTL.

TIME_CODE_MODULUS 30
RECORD_FRAME_RATE:          29.97
PRODUCTION_TITLE:           Alpo New//Improved MIS
PRODUCTION_NUMBER:          BBDO 910519-100
SHOOT_DATE:                 05/23/91

TRANSFER_FACILITY:          Budget Kinescope, Ltd.
TRANSFER_DATE:              05/21/91
OPERATOR:                   Jean-Paul LaFoote (and his dog, Spot)
TRANSFER_BAY:               Suite A   9:00 to 15:30

RECORD B A-ROLL 01.00.00.00
RECORD B B-ROLL 01:00:00.00
FILM_FORMAT:                ROLL01  35MM/4
SOURCE_MEDIUM:              ROLL01  5297
FRAME_RATE:                 ROLL01  24  23.98
EDGE_NUMBER_TYPE:           ROLL01  BAR
EDGE_NUMBER_PERIOD:         ROLL01  64/4

001  ROLL01       V   T   KJ5643   1008+00A   00:02:00:00   09:15
NORECORD B B-ROLL
SOURCE_MEDIUM:              ROLL02  5297
FILM_FORMAT:                ROLL02  35MM/4
FRAME_RATE:                 ROLL02  24  23.98
EDGE_NUMBER_TYPE:           ROLL02  BAR
EDGE_NUMBER_PERIOD:         ROLL02  64/4

SOURCE_MEDIUM:              REEL01  Sync Audio  1/4”  CT  TC  15 in/s
FRAME_RATE:                 REEL01  30  29.97

SCENE: 1
TAKE: 3
SCRIPT: "ANCR: We’re here today to see how the man in the street..."
002   ROLL02       V  T    KJ5640   1006+15C   00:02:09:15   02:15
002   REEL01.1-2   A  T    09:55:16:02            00:02:09:15  02:15

SCENE: 2
TAKE: 7
SCRIPT: "MIS:  What’s new about it?"
003   ROLL02       V  T    KJ5640 1156+23A     00:02:12:00   2:06
003   REEL01.1-2   A  T    09:95:00:19          00:02:12:00  2:06

SCENE: 2
TAKE: 9
004   ROLL02       V  T    KJ5640 1180+10A     00:02:14:06   4:15
004   REEL01.1-2   A  T   10:05:19:19          00:02:14:06   4:15
```

NOTES

1 The "TRANSFER_BAY:" data item is defined by the maker of this list to document some information not predefined in this practice.

2 A record medium is defined using the identifier "A-ROLL" which is recording in a video system with a frame rate of 29.97 with the video and all audio channels active. All video time codes in transfer elements have an actual offset of one hour to the time code used on this medium.

3 Another record medium ("B-ROLL") is also made active.

4 The film format (there is only one at a time) is 35-mm 4 perf.

5 A source medium is defined and named "ROLL01" and is noted to be Eastman 5297 color negative. The transfer system may attempt to interpret the string after the roll identifier and make use of this information.

6 The frame rate of this film when it was exposed was 24 frames/s, but it is being transferred locked to the video rate of the system for an effective rate of 23.98 frames/s.

7 Transfer number 1 is done to both the A-ROLL and B-ROLL with the film frame adjacent to the latent image bar code number transferred to the video as an 'A' film frame. The duration of the transfer is 228 film frames. The duration of the transfer on the output video is 285 video frames.

8 The ABCD sequence is not preserved by this transfer; it is interrupted at the start of transfer numbers 2 and 4.

Annex C (informative)
Bibliography

ANSI/SMPTE 125M-1995, Television — Component Video Signal 4:2:2 — Bit-Parallel Digital Interface

SMPTE 96-1999, Television — 35- and 16-mm Motion-Picture Film — Scanned Image Area

SMPTE RP 195-1998, Use of the Reference Mark in Manufacturer-Printed Latent Image Key Numbers for Unambiguous Film Frame Identification

EBU Tech. 3245 (1988), Remote-Control Systems for Broadcasting Production Equipment (ESbus), Supplement 4 (1989): Telecine Type-Specific Messages

EBU Tech. 3267 (1992), EBU Interfaces for 625-Line Digital Video Signals at the 4:2:2 Level of CCIR Recomendation 601